

```

// Commande principale: // démarrer
pour nouveau
// initialise les paramètres par défaut de l'écran,
// du crayon et de la tortue. //
eff déroule fcc verte trait 1 cachetortue
fin

pour spire
donne "l xycurve 0 1
donne "x1 prem :l donne "y1 der :l
répète 360 [
fcc angcol boucle
donne "l xycurve boucle :cas //:cas = 1, 2 ou 3 //
donne "x prem :l donne "y der :l
segment ph :x1 :y1 ph :x :y
donne "x1 :x
donne "y1 :y
]
fin

pour aléa
fixehasard 1 5
donne "r item [80 90 100 110 120] hasard
donne "s 160 -:r fixehasard 0 9
donne "a (1 + hasard )
donne "b (1 + hasard )
donne "c (1 + hasard )
donne "d (1 + hasard )
fin

pour angcol :theta
fixehasard 1000 32500
rends hasard
fin

pour xycurve :t :cas //:cas = 1, 2 ou 3 //
teste :cas =1
sivrai [
donne "x :r *(sin :a *:t) + :s *(sin :b *:t)
donne "y :r *(cos :c *:t) + :s *(cos :d *:t)
rends ph :x :y
]
sifaux [
teste :cas = 2
sivrai [ // spires //
donne "x :r *(cos :a *:t) + :s *(sin :b *:t)
donne "y :r *(sin :c *:t) + :s *(cos :d *:t)
rends ph :x :y
]
sifaux [ // fleurs (note: les paramètres :c et :d non utilisées.//
donne "x :r *(cos :a *:t) + :s *(sin :b *:t)
donne "y :r *(sin :a *:t) + :s *(cos :b *:t)
rends ph :x :y
]
]
fin

pour démarrer
partage [ "a "b "c "d "r "s "t "cas]
effxt donne "cas 1
répète 3 [
ecl :cas
répète 12 [nouveau aléa spire attends 0,25]
donne "cas :cas +1
]
fin

```